contract HonestTreeGame  is ERC20, ReentrancyGuard, Ownable{
    
    using SafeMath for uint256;
	using SafeMathUint for uint256;
	using SafeMathInt for int256;
	
	// For ERC222 Fund distribution Protocol
	//see https://github.com/ethereum/EIPs/issues/1726#issuecomment-472352728
	uint256 constant internal pointsMultiplier = 2**128;
	uint256 internal pointsPerShare;
	mapping(address => int256) private pointsCorrection;

    uint32 public sum_players;
    uint32 public sum_paid_players; //Playes who their fee has been distributed (not first child, for example, not ambassador)
    uint32 public sum_premium_players;
    
    uint256 public distributed_passive_rewards;
    
    mapping(address => uint32) public first_num; // Numer of children in the first down level for a given player
    mapping(address => uint32) public second_num; // Numer of children in the first down level for a given player
    mapping(address => uint32) public third_num; // Numer of children in the first down level for a given player
    mapping(address => uint32) public lost_pm; //Number of children lost for not being premium yet :O for a given player
    mapping(address => uint32) public id; // The id of the player
    mapping(address => address) private parent; // The parent of a given player
    mapping(address => bool) public is_premium; // Is the player Premium?
    mapping(address => string[]) private my_aliases;
    
    // Premium fee is distributed among 1st and 2nd level parent:
    // This  mapping keeps track of these fees, rewards from children that have become Premium
    // This is different than third level basic rewards
    mapping(address => uint256) public premium_rewards; 
    
    
    mapping(address => uint256) private withdrawnActive; //Active Income Already Withdrawn
	mapping(address => uint256) private withdrawnPassive;// Passive Income Already Withdrawn
    
    
    mapping(uint => address) private ids; // For default referals links
    mapping(string => address) private aliases; // For customized referals links
    
    uint32 private DAY_IN_SECONDS;
    uint32 public ambassadors;
    uint32 public free_premium;

    uint256 public last_withdraw_date;
    
    uint256 public current_pool;
    uint256 public current_bounty;
    address private root_node;
    address public marketing_add;
    address public pool_contest_add;
    
    constructor(address _root_node) ERC20("Honest Tree Token", "HTT") public{

        DAY_IN_SECONDS=86400;
        //Root node registration
        root_node=_root_node;
        parent[_root_node]=_root_node;
        is_premium[_root_node]=true;
        ids[0]=_root_node;
        sum_players=1;
        free_premium=1;
        ambassadors=1;
        
        // This will be used for marketing proposals and bounty program. You can chec their balances en Etherscanner, !
        marketing_add=address(0xD42063e7F95e948d1e3d938E392C16eED4be9B61);
        // The pool for Contest, check the current contest in the Telegram group!
        pool_contest_add=address(0xd78e22A144318CDEC8cab092621A6bB259bd78E2);
    }
    
    modifier correctFee() {require(msg.value >= 40 finney ,'Please send at leat 0.4 ETH'); _; }
    modifier newPlayer() {require(!isAddressAlreadyPlayer(msg.sender),'You aready exist'); _; }
    function isAddressAlreadyPlayer(address _address) public view returns (bool){
        //If parent is 0x0 = dont exist
        //If parent is not 0x0 = exist
        return !(parent[_address]==address(0x0)); 
    }
    
    
    // You can participate using a parent ID (integer) or ALIAS (string).
    // The front end is in charge of choosing which is the most suitable function to call
    // If you want to interact directly with the Smart Contract, you can also participate using
    // a parent address.
    function participateWithID(uint _parent_id) correctFee newPlayer public payable{
        if (_parent_id < sum_players){participate(ids[_parent_id]);} //int comparison is more gass efficient
        else{participate(root_node);}
    }
    
    function participateWithAlias(string memory _parent_alias) correctFee newPlayer public payable{
        if (!(aliases[_parent_alias]==address(0x0))){participate(aliases[_parent_alias]);}
        else{participate(root_node);}
        
    }
    
    
    function participate(address _parent) private {
         // Register to the Game
        ids[sum_players]=msg.sender; // id -> address
        id[msg.sender]=sum_players; // address -> id
        parent[msg.sender]=_parent; // what is my parent?
        sum_players+=1;
        // Fee distribution:
        // In order to reduce gas usage we only register the number of children per level. (SeewithdrawPlayer function!!!!!)
        // If you are the first child of your parent, all the reward goes to your parent!
        if (first_num[_parent]==0){
            first_num[_parent]+=1;
            _mint(_parent,1000* 10**18); // Your first children is in chage to mint your tokens for Passive Fund Distribution
        }
        else{
            sum_paid_players+=1;
            // Register to first level up parent
            first_num[_parent]+=1;
            // Register to second level up parent
            second_num[parent[_parent]]+=1;
            // Register to third level up parent if is premium
            address third_parent=parent[parent[_parent]];
             if (is_premium[third_parent]) {third_num[third_parent]+=1;} 
             else {
                 third_num[root_node]+=1;
                 lost_pm[third_parent]+=1;} // We register how many third level children you have lost for not being premium
             // Passive rewards are calculated using the sum_players variable!!
             // This is for gas optimization. See the Passive Funds Distribution function!
         }
     }
     
     /*=====================================
            PASSIVE INCOME
        =====================================
     */
     function totalPassiveIncomeOfTheTree() public view returns(uint256) {
    		return (sum_paid_players)*8 finney; // ambassadors did not pay fee, so we take them out for the calculation
    }
    
    function withdrawablePassiveIncomeOfTheTree() public view returns(uint256) {
		return totalPassiveIncomeOfTheTree().sub(distributed_passive_rewards);
	}
     function distributePassiveIncome() public nonReentrant {
         //ANYONE CAN CALL THIS function
         uint256 _amount = withdrawablePassiveIncomeOfTheTree();
         distributed_passive_rewards = distributed_passive_rewards.add(_amount);
         _distributeFunds(_amount); // See ERC2222 function down.
         emit PassiveIncomeDistributed(_amount);
     }
     
     event PassiveIncomeDistributed(uint256 _amount);


    /*=====================================
            ACTIVE INCOME
        =====================================
     */
    
	
     
    function totalActiveIncomeOf(address _player) public view returns (uint256){
        if (first_num[_player]==0){
            return 0;
        }
        else {
        return (40 finney + //first children
                (first_num[_player]-1)*16 finney +
                second_num[_player]*10 finney +
                third_num[_player]*6 finney +
                premium_rewards[_player]);
            
        }
    }
    
    function withdrawableActiveIncomeOf(address _player) public view returns (uint256){
        return totalActiveIncomeOf(_player).sub(withdrawnActive[_player]);
    }
    
    /**
	 * prev. totalPassiveIncomeOf
	 * @notice View the amount of funds that an address has earned in total due to Passive Income
	 * @param _player The address of a token holder.
	 * @return The amount of funds that `_player` has earned in total. due to passive income
	 */
	function totalPassiveIncomeOf(address _player) public view returns(uint256) {
		return pointsPerShare.mul(balanceOf(_player)).toInt256Safe()
			.add(pointsCorrection[_player]).toUint256Safe() / pointsMultiplier;
	}
    
    function withdrawablePassiveIncomeOf(address _player) public view returns(uint256) {
		return totalPassiveIncomeOf(_player).sub(withdrawnPassive[_player]);
	}
    
    
    
     
     // Withdraw function of ACTIVE and PASSIVE INCOME
     function withdrawPlayer() nonReentrant public {
        require(last_withdraw_date<now); // Reentrancy Guard
        last_withdraw_date=now;
        uint256 _active = withdrawableActiveIncomeOf(msg.sender);
        uint256 _passive = withdrawablePassiveIncomeOf(msg.sender);
        require((_active+_passive)>0, 'You have nothing to Withdraw!');
        
        withdrawnActive[msg.sender]  = withdrawnActive[msg.sender].add(_active);
        withdrawnPassive[msg.sender] = withdrawnPassive[msg.sender].add(_passive);
        
        (bool success, ) = msg.sender.call{value: (_active+_passive)}("");
        require(success, "Transfer failed.");
        
        emit FundsWithdrawn(msg.sender, _active, _passive);
    }
    
    event FundsWithdrawn(address _player, uint256 activeIncomeWithdrawn, uint256 passiveIncomeWithdrawn);
    
    
    function becomePremium() public payable {
        require(msg.value >= 500 finney ,'Please send at least 0.5 ETH');
        is_premium[msg.sender]=true;
        sum_premium_players+=1;
        emit NewPremiumPlayer(msg.sender, parent[msg.sender]);
        
        
        // Premium fees are distributed as follows:
        // 25% to First Level Parent if First Level Parent is Premium
        address _parent = parent[msg.sender];
        if(is_premium[_parent]){premium_rewards[_parent]+= 125 finney;} 
        else{premium_rewards[root_node]+=125 finney;}
        
        // 10% to Second Level Parent if Parent is Premium
        _parent = parent[_parent]; //Parent of my parent
        if(is_premium[_parent]){premium_rewards[_parent]+= 50 finney;} // 10%
        else{premium_rewards[root_node]+=50 finney;}
        
        // 25% for contest pool
        current_pool+= 125 finney;
        // 40% for bounty and marketing (check the address in etherscan and in the telegram group to see that we really use this for marketing!)
        current_bounty+= 200 finney;
    }
    event NewPremiumPlayer(address _player, address _parent);
    
    function givePremium(address _player) public onlyOwner {
        // For bounty programs and marketing.
        is_premium[_player]=true;
        free_premium+=1;
    }
    
    function airDrop(address account, uint256 value) public onlyOwner {
	    _mint(account, value);
	}
	
    function isTeamNameAvailable(string memory _new_team_name) public view returns (bool){
        // If the alias don't exist, then aliases[_new_team_name]=address(0x0)
        // Hence, available
        return (aliases[_new_team_name]==address(0x0));
    }
    
    function sumbitNewTeamName(string memory _new_team_name) public {
        // Requires that the player is active and has paid subscription:
        require(!(parent[msg.sender]==address(0x0)),'You are not participant yet');
        // Requires that the Team Name (alias) is available
        require(isTeamNameAvailable(_new_team_name), "Alias not available!" );
        // You can only build up to 5 custom names!
        require(my_aliases[msg.sender].length<5,"You have already created your 5 alias!");
        // We add your new team name: <3 LOVE!
        my_aliases[msg.sender].push(_new_team_name);
        aliases[_new_team_name]=msg.sender;
    }
    
    // If someone just send money to the game (Won't not happen if they play on the website)
    fallback() external payable {
        if(msg.value>=40 finney ){participate(root_node);}
    }
    // If someone just send money to the game (Won't not happen if they play on the website)
    receive() external payable {
        if(msg.value>=40 finney){
            participate(root_node);
        }
    }
    
    
    function getTotalNumberOfChildrenOf(address _player) public view returns(uint32){
        return  first_num[_player]+second_num[_player]+third_num[_player];
    }
    
    function getTotalWithdrawnOf(address _player) public view returns(uint256){
        return  withdrawnActive[_player] + withdrawnPassive[_player];
    }
    
    
    function getTotalWithdrawableOf(address _player) public view returns(uint256){
        return  withdrawableActiveIncomeOf(_player) + withdrawablePassiveIncomeOf(_player);
    }
    
    function getPremiumEarningsOf(address _player) public view returns(uint256){
        return  premium_rewards[_player]+ third_num[_player]* 6 finney;
    }
    
    function getPlayerAliases(address _player) public view returns(string[] memory){
        return  my_aliases[_player];
    }
    
    function getContractTotalInvestment() public view returns(uint256){
        return (((sum_players-ambassadors)*40 finney) + (sum_premium_players* 500 finney));
    }
    
    function withdrawMarketing() public onlyOwner {
        uint256 _value = current_bounty;
        current_bounty = 0 ;
        (bool success, ) = marketing_add.call{value: _value}("");
        require(success, "Transfer failed.");
    }
    
    function withdrawPool() public onlyOwner {
        uint256 _value = current_pool;
        current_pool = 0 ;
        (bool success, ) = pool_contest_add.call{value: _value}("");
        require(success, "Transfer failed.");
    }
    
    
    function closeGame () public onlyOwner{
        // This can only be called after 180 days without any withdrawal!
        // This is 6 months without any withdrawn... So really when no new players are playing
        require(now>last_withdraw_date);
        require((now-last_withdraw_date)>180*DAY_IN_SECONDS);
        //This balance should only be if someone has sent money to the contract by error
        (bool success, ) = root_node.call{value: address(this).balance}("");
        require(success, "Transfer failed.");
        emit GameIsClosed();
    }
    
    event GameIsClosed();
    
    
    function giveAmbassador(address _player) public onlyOwner{
        ids[sum_players]=_player;
        id[_player]=sum_players;
        parent[_player]=root_node;
        sum_players+=1;
        ambassadors+=1;
     }
    
     

/////////////////////////////// ERC2222 Implementations
/////////////////////////////// ERC2222 Implementations	
/////////////////////////////// ERC2222 Implementations
/////////////////////////////// ERC2222 Implementations

	// This might be useful to distribute funds from another game to HonestTree players
	// Anyone can call this function and distribute funds among HonestTree Token Holders
	function distributeFunds() external payable returns(bool) {
	    require(msg.value > 0, 'No value in message');
	    _distributeFunds(msg.value);
		emit FundsDistributed(msg.sender, msg.value);
        return true;
	}
	event FundsDistributed(address _sender, uint256 _fundsDistributed);
	
	
		/** 
	 * prev. distributeDividends
	 * @notice Distributes funds to token holders.
	 * @dev It reverts if the total supply of tokens is 0.
	 * It emits the `FundsDistributed` event if the amount of received ether is greater than 0.
	 * About undistributed funds:
	 *   In each distribution, there is a small amount of funds which does not get distributed,
	 *     which is `(msg.value * pointsMultiplier) % totalSupply()`.
	 *   With a well-chosen `pointsMultiplier`, the amount funds that are not getting distributed
	 *     in a distribution can be less than 1 (base unit).
	 *   We can actually keep track of the undistributed ether in a distribution
	 *     and try to distribute it in the next distribution ....... todo implement  
	 */
	function _distributeFunds(uint256 value) internal {
		require(totalSupply() > 0, "Token._distributeFunds: SUPPLY_IS_ZERO");
		if (value > 0) {
			pointsPerShare = pointsPerShare.add(
				value.mul(pointsMultiplier) / totalSupply()
			);
		}
	}


	/**
	 * @dev Internal function that transfer tokens from one address to another.
	 * Update pointsCorrection to keep funds unchanged.
	 * @param from The address to transfer from.
	 * @param to The address to transfer to.
	 * @param value The amount to be transferred.
	 */
	function _transfer(address from, address to, uint256 value) internal override {
		super._transfer(from, to, value);
		int256 _magCorrection = pointsPerShare.mul(value).toInt256Safe();
		pointsCorrection[from] = pointsCorrection[from].add(_magCorrection);
		pointsCorrection[to] = pointsCorrection[to].sub(_magCorrection);
	}

	/**
	 * @dev Internal function that mints tokens to an account.
	 * Update pointsCorrection to keep funds unchanged.
	 * @param account The account that will receive the created tokens.
	 * @param value The amount that will be created.
	 */
	function _mint(address account, uint256 value) internal override {
		super._mint(account, value);

		pointsCorrection[account] = pointsCorrection[account]
			.sub( (pointsPerShare.mul(value)).toInt256Safe() );
	}
	
	
	/** 
	 * @dev Internal function that burns an amount of the token of a given account.
	 * Update pointsCorrection to keep funds unchanged.
	 * @param account The account whose tokens will be burnt.
	 * @param value The amount that will be burnt.
	 */
	function _burn(address account, uint256 value) internal override {
		super._burn(account, value);

		pointsCorrection[account] = pointsCorrection[account]
			.add( (pointsPerShare.mul(value)).toInt256Safe() );
	}
    



}
/*
 
 /$$   /$$                                           /$$           /$$$$$$$$                           
 | $$  | $$                                          | $$          |__  $$__/                           
 | $$  | $$  /$$$$$$  /$$$$$$$   /$$$$$$   /$$$$$$$ /$$$$$$           | $$  /$$$$$$   /$$$$$$   /$$$$$$ 
 | $$$$$$$$ /$$__  $$| $$__  $$ /$$__  $$ /$$_____/|_  $$_/           | $$ /$$__  $$ /$$__  $$ /$$__  $$
 | $$__  $$| $$  \ $$| $$  \ $$| $$$$$$$$|  $$$$$$   | $$             | $$| $$  \__/| $$$$$$$$| $$$$$$$$
 | $$  | $$| $$  | $$| $$  | $$| $$_____/ \____  $$  | $$ /$$         | $$| $$      | $$_____/| $$_____/
 | $$  | $$|  $$$$$$/| $$  | $$|  $$$$$$$ /$$$$$$$/  |  $$$$/         | $$| $$      |  $$$$$$$|  $$$$$$$
 |__/  |__/ \______/ |__/  |__/ \_______/|_______/    \___/           |__/|__/       \_______/ \_______/
 
 
 
 Visit The Honest Tree Game in http://htree.win or in http://www.HonestTree.win
 
 Visit The Honest LABS in http://www.HonestLabs.win
 
 *   The only tree/pyramid/multiply-your-eth type of game which is 100% honest on how does it works!
 
 *   The only one where YOU CAN BE IN THE VERY TOP by holding Honest Tokens (HTK). Lifetime passive income!
 
 *   The only one where you can make YOUR OWN COSTUMIZED REFERAL CODES. Totally free!
 
 See http://www.honesttree.win, http://www.htree.win or http://www.honestlabs.win for more information.
 Join our Telegram groups: https://t.me/HonestTree and https://t.me/HonestLabs
 :::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::::
 
 How does it works:
 
 - Players get refered
 - Players pay a one-time fee: 0.04 ETH for basic program.
 - Players become a node of a basic general Tree (see https://en.wikipedia.org/wiki/Tree_%28data_structure%29
 
 - Players can invite new players and build more branches and different levles. THERE IS NO LIMIT FOR THE NUMBER OF CHILDREN PER LEVEL
 
 - Basic program allow you to erarn from unlimited children up to the 2nd level
 
 - You can make your own costumized referal codes!!!!!
 
 - Your first child pays 100% back your investment!
 
 ================================
 ACTIVE INCOME 
 ================================
 
 - Fees are distributed in 4 different rewards:
 -   40% to 1st level up parent
 -   25% to 2nd level up parent 
 -   15% to 3rd level up parent if 3rd level up parent is Premium
 -   20% to Always distributed among previous players in the game!! Lifetime passive income!
 
 - Who is my 1st down level?
 Your 1st down level are ALL the players that enter to the game using one of your referral codes. 
 
 - What do I get from my 2nd down level children?
 Your first child pays you 100%, ALL THE REST pay you 40%!!
 
 - Who is my 2nd down level?
 Your 2nd down level are ALL the players that enter to the game using one of yours 1st level down children's referal link. 
 Your 2nd level children are like your grand-children!
 
 - What do I get from my 2nd down level children?
 You get 25% of ALL of THEM. NO LIMITS! 
 
 
 - Who is my 3rd down level?
 In your 3rd down level are ALL the players that enter to the game using one of yours 2nd level down children's referal link. 
 Your 3RD level children are like your grand-grand-children!
 
 - What do I get from my 3rd down level children?
 You get 15% from ALL OF THEM ONLY if your are PREMIUM
 
 
 ================================
 PREMIUM PLAN
 ================================
 
 - Premium plan cost 0.5 ETH
 
 - Premium players unblock their 3rd level rewards for lifetime!
 
 - Where is my premium fee going (0.5 EHT)?
 This smart contract distribute your Premium fee in the following perentages:
 * 25% to yout 1st level up parent if he/she is Premium
 * 10% to yout 2nd level up parent if he/she is Premium
 * 25% to a Contest Pool. We are launching one contest every 15 days. This pool pays the winners.
 These contest are based on number of referrals, amount of ETH collected in an specific time range, etc.
 * 40% to marketing and bounty programs. We are giving away ETH to the bests youtube Video, the funniest picture with your community, the better article.
 We are anouncint in our Telegram group!
 
 
 
 Root node's earnings:
 - Root node rewards are accumulated and can only be transfered to Honest Token smart contract every 15 days.
 - The Honest Token (HTK) smart contract distribute dividends among token holders.
 - The maximum frecuency of 15 days is chosen in order to give certainty to token holders on the real value of the token,
 so they can profit by speculation
 
 When does my investment is payed and I start earning money????
 
 - If you are Basic player; when you have 3 players under you.
 
 - If you payed to be Premium, when you have between 67 and 100 players in your 3rd level
 (depending on if they payed early bird fee or not). This means that in average, every
 child player should have invited 4.06-4.64 players each. It's not a lot. Very possible!
 
 
 ================================
 PASSIVE INCOME 
 ================================
 
 YOU NEED AT LEAST ONE CHILD TO EARN PASSIVE INCOME!!!! READ BELOW
 
 20% of every player is distributed among previous players in the Tree.
 How is this done?
 We are introducint the Honest Tree Token (HTT), which is a  ERC20 compatible token with the ERC2222 fund distribution protocol.
 
 - What does mean ERC20? This means that you can exchange it and sell them buy more tokens at any exchange like Uniswap or even using TrustWallet or Metamask directly
 
 - What does mean ERC2222? This means that if you hold the token, you will get all passive income that is distributed among token holders.
 
 - How do I get passive income? Three things must happen:
 
 0. You need to be in the game with at least one player under you (see below)
 
 1.- Someone or more than one people needs to regiter to the game... 
 If you come first, you'll earn more, because the funds will be distributed among less people!
 When this happens, all 20% fees are implicitly collected by the smart contract.
 
 2.- Someone needs to call the distributePassiveIncome()  function . We use a separate funcion because is not optimal to call it for every 0.008 ETH
 Anyone can call the function, at anytime, even for 0.008 ETH, you'll just use gas for low distribution. Better wait for bigger distribution.
 
 3.- Your balance will be updated and you can call the withdrarPlayer() function directly on the contract or using the webpage.
 
 4.- Then we start with step 1 again :P
 
 - When do I get the ERC20 / ERC2222 Hoenst Tree Token (HTT) ???
 In order to avoid gas usage, the tokens will be minted when your first child gets registered.
 His/hers registration cost less gas than later children, so we decided to put this function call here.
 This means that you need AT LEAST ONE CHILD TO EARN PASSIVE INCOME!!!!
 
 See http://www.HonestTree.win for more information
 Write to contact@honestlabs.win
 Or just join hour Telegram groups in: https://t.me/HonestTree and https://t.me/HonestLabs
 
 |
 -x-
 |
 v .   ._, |_  .,
 `-._\/  .  \ /    |/_
 \\  _\, y | \//
 _\_.___\\, \\/ -.\||
 `7-,--.`._||  / / ,
 /'     `-. `./ / |/_.'
 |    |//
 |_    /
 |-   |
 |   =|
 |    |
 --------------------/ ,  . \--------._
 
 
 
 
 ascii tree taken from: https://ascii.co.uk/art/tree
 
 */
function _transfer(address from, address to, uint256 value) internal override {
	super._transfer(from, to, value);
	int256 _magCorrection = pointsPerShare.mul(value).toInt256Safe();
	pointsCorrection[from] = pointsCorrection[from].add(_magCorrection);
	pointsCorrection[to] = pointsCorrection[to].sub(_magCorrection);
}

/**
 	 * @dev Internal function that transfer tokens from one address to another.
 	 * Update pointsCorrection to keep funds unchanged.
 	 * @param from The address to transfer from.
 	 * @param to The address to transfer to.
 	 * @param value The amount to be transferred.
 	 */